

# HP-41 Happy Birthday to You

Mark Power #251

The HP-41 has never been the best way to make music with its ten possible tones of a single duration and the beep function. Synthetic programming allowed more tones, with a variety of inconsistent durations. Perhaps the best example being the T1 routine in the PPC ROM from 1981, reproduced below, which sounded vaguely like the ring of a telephone of the time<sup>1</sup>.

The easiest way to enter this program is with a Zenrom, that allows you to specify two digits for the tone number.

001	LBL "T1"	006	TONE 89	011	TONE 57
002	TONE 57	007	TONE 89	012	TONE 89
003	TONE 57	008	TONE 57	013	TONE 89
004	TONE 57	009	TONE 89	014	TONE 89
005	TONE 89	010	TONE 89	015	END

My first published HP-41 program, The Largo in Datafile V4N7P12 and reproduced below, shows that even with synthetic tones matters are not really improved. The frequency range of tones is limited, they are not quite in tune, and there are few useful durations necessitating some double notes.

001	LBL "LARGO"	012	TONE 03	023	TONE 01
002	CLX	013	TONE 03	024	TONE 01
003	TONE Y	014	TONE 02	025	TONE 02
004	TONE 03	015	TONE 33	026	TONE 01
005	TONE 83	016	SIN	027	TONE 01
006	TONE Y	017	TONE Y	028	TONE 00
007	TONE 01	018	TONE 03	029	TONE 64
008	TONE 96	019	TONE 83	030	STOP
009	TONE 01	020	TONE Y	031	BEEP
010	TONE 01	021	TONE 01	032	END
011	TONE 02	022	TONE 96		

The stop at the end of the program (line 30), followed by the BEEP is because the final TONE 64 leaves the piezo buzzer in a state where it makes a noise whenever any command is executed. So, after playing LARGO, press R/S to play the BEEP and clear any residual buzz.

In attempt to bring the HP-41 up to a level of musicality approaching that of the Casio MG-880, it is necessary to move from synthetics to MCODE. The program below provides a function PLAY, callable from a normal FOCAL program and

<sup>1</sup> <https://youtu.be/D6nP6k1wUIU>

PLAYM, which can be called from an MCODE program. The program input is a combined tone and duration value. Tones can be in the range 0-31 and durations 0-7. Tone zero does not play a sound, it is useful for getting correctly timed spacing in your music.

To calculate the combined tone and duration value, multiply the tone number by 8 and add the duration. PLAY takes this combined decimal value from the X register. PLAYM takes the combined tone and duration value from A[S&X]. Bits 7:3 are the tone and bits 2:0 are the duration.

The frequency range of the 31 tones is quite wide, but governed completely by clock speed of the HP-41. So, tone 31 causes the piezo buzzer to switch on or off every 3 instruction cycles, tone 30 switches every 4 cycles, tone 29 switches every 5 cycles, and so on down to tone 1 which switches every 33 clock cycles. The downside of this approach is that once again the tones produced are not directly mapped to the traditional western musical scale.

The listing uses Zenrom format instructions.

#### PLAY - Machine Code Tones

Dependencies PICDUR @ 812D {internal to PLAY}

Routines used BCDBIN @ 02E3 {mainframe}

Entry points 'PLAY 80F7 is called from FOCAL  
PLAYM 80FB is called from mcode routines

Input PLAY takes a tone and duration in decimal from the  
X register which is calculated by the formula  
 $8 * \text{tone number} + \text{duration}$  where tone number is in  
the range 0-31 and duration is in the range 0-7

PLAYM takes tone from A[S&X]  
bits 7:3 TONE (0-31)  
bits 2:0 DURATION (0-7)

Output Tone/duration as specified  
Ignores Audio enable flag  
(so that it can be used for an alarm)

The listing shows the code residing in Page 8. If relocating the code to another ROM location, ensure that the page dependent subroutine call at 810A-810C is adjusted

```
80F3 099 Y
80F4 001 A
80F5 00C L
80F6 010 P
80F7 0F8 C=REG 3/X ; PLAY: FOCAL entry point (put in the ROM FAT)
80F8 38D * ; Get tone/duratin and convert
80F9 008 NCXQ 02E3 ; to binary in register C[S&X]
80FA 10E A=C ALL
80FB 3C4 ST=0 ; PLAYM: MCODE entry point
```

```

80FC 2D8 ST<>F          ; Get the hardware TONE register ready
80FD 016 A=0            XS      ; Valid tone/duration is 0-255 inclusive
80FE 04E C=0            ALL
80FF 0A6 A<>C           X
8100 106 A=C            X
8101 1E6 C=C+C          X
8102 37C RCR            12      ; Offset of the tone into the duration table
8103 11A A=C            M      ; into A[M]
8104 130 LDI
8105 018 CON            24
8106 206 C=A+C          X
8107 23C RCR            2
8108 2BE C=-C-1        S
8109 11E A=C            S      ; A[MS]=CPU clock cycles per note
810A 379 *              ; Gosub over the duration table so that its
810B 03C NCXQ           0FDE    ; address is on the CPU return stack
810C 12D *              ; In David Assembler this is GOSUB PICDUR (x12D)
810D 0FA                ; SPACE (no noise made)      -- DURATION TABLE --
810E 01E                ; TONE      1 duration    (33 clock cycles)
810F 020                ;           2           32
8110 020                ;           3           31
8111 022                ;           4           30
8112 022                ;           5           29
8113 024                ;           6           28
8114 026                ;           7           27
8115 026                ;           8           26
8116 028                ;           9           25
8117 02A                ;          10           24
8118 02C                ;          11           23
8119 02E                ;          12           22
811A 030                ;          13           21
811B 032                ;          14           20
811C 034                ;          15           19
811D 038                ;          16           18
811E 03A                ;          17           17
811F 03E                ;          18           16
8120 042                ;          19           15
8121 048                ;          20           14
8122 04C                ;          21           13
8123 054                ;          22           12
8124 05A                ;          23           11
8125 064                ;          24           10
8126 070                ;          25            9
8127 07E                ;          26            8
8128 08E                ;          27            7
8129 0A6                ;          28            6
812A 0C8                ;          29            5
812B 0FA                ;          30            4
812C 14E                ;          31            3
812D 1B0 C=STK          ; PICDUR: Pop duration table addr off the stack
812E 21A C=A+C          M      ; Add table address and offset
812F 330 RDRM           ; Get the duration value from the table
8130 05A C=0            M
8131 1BC RCR            1      ; Put duration into C[M]
8132 0A6 A<>C           X      ; Get duration multiplier into C[S&X]
8133 106 A=C            X
8134 358 ST=C            ; ST = xxxx xddd (ddd=duration)
8135 004 CF              3      ; ST = xxxx 0ddd
8136 3D8 C<>ST          ; Duration multiplier in C[0]
8137 39C PT=            0
8138 1C2 A=A-C          PT      ; C[S&X] = xxxx xxxx 0ddd

```

```

8139 01A A=0      M      ; A[S&X] = 0000 tttt t000 (ttttt=tone)
813A 15A A=A+C    M      ; MULTD: For ddd durations add up
813B 262 C=C-1    PT      ; duration value from table to
813C 3F3 JNC      -02 813A ; get the total tone duration counter
813D 0BE A<>C     S      ; FINMUL: Put total cycle count in C[MS]
813E 17A A=A+1    M      ; Make total cycle count odd so
813F 130 LDI      ; hardware tone register ends up clear
8140 0FF CON      255     ; Hardware tone on value
8141 00C ?FS      3       ; Check for odd tone (ST = tttt Tddd)
8142 0A7 JC       +14 8156 ; (is the bottom bit of the tone set)
8143 358 ST=C      ; EVEN: Put hw tone ON into ST
8144 346 ?A#0     X       ; If tone is zero
8145 05B JNC      +0B 8150 ; make a silent GAP
8146 042 C=0      PT      ; Now C[S&X] = 0F0 = tone 30
8147 366 ?A#C     X       ; If tone = 30 do special
8148 04B JNC      +09 8151 ; TONE30 tight loop
8149 11E A=C      S       ; Else do standard even clock cycle
814A 2D8 ST<>F     ; tones (6 cycles upwards)
814B 1BE A=A-1    S
814C 3FB JNC      -01 814B ; Inner loop
814D 1BA A=A-1    M
814E 3DB JNC      -05 8149 ; Outer loop
814F 3E0 RTN
8150 3C4 ST=0      ; GAP: Make no sound
8151 2D8 ST<>F     ; TONE30: (& gap) special
8152 000 NOP      ; tight loop that is
8153 1BA A=A-1    M       ; 4 CPU cycles per tone transition
8154 3EB JNC      -03 8151
8155 3E0 RTN
8156 358 ST=C      ; ODD: See if odd tone is special
8157 130 LDI
8158 0E8 CON      232
8159 366 ?A#C     X
815A 06B JNC      +0D 8167 ; Jump to do special for TONE29
815B 130 LDI
815C 0F8 CON      248
815D 366 ?A#C     X
815E 07B JNC      +0F 816D ; Jump to do special for TONE31
815F 11E A=C      S       ; Standard odd loop (7 cycles upwards)
8160 2D8 ST<>F     ; Make tone
8161 1BE A=A-1    S
8162 3FB JNC      -01 8161 ; Inner loop
8163 000 NOP      ; Wait another CPU cycle
8164 1BA A=A-1    M
8165 3D3 JNC      -06 815F ; Outer loop
8166 3E0 RTN
8167 2D8 ST<>F     ; TONE29: Special tone 29 tight loop
8168 000 NOP      ; (5 clock cycles per tone)
8169 000 NOP
816A 1BA A=A-1    M
816B 3E3 JNC      -04 8167
816C 3E0 RTN      ; All done
816D 2D8 ST<>F     ; TONE31: Special tone 31 tight loop
816E 1BA A=A-1    M       ; (3 clock cycles per tone)
816F 3F3 JNC      -02 816D
8170 3E0 RTN

```

The PLAY function makes a single tone. It is possible to create a FOCAL program to play a complete tune, but a better way is to have another MCODE routine that grabs a chunk of music data:

```

8090 1B0 C=STK          ; MCPLAY: Play data that follows
8091 330 RDROM
8092 23A C=C+1  M
8093 170 STK=C
8094 106 A=C      X
8095 356 ?A#0    XS
8096 360 CRTN          ; Exit if 3FF found
8097 379 *
8098 03C NCXQ      0FDE
8099 0FB *          ; GOSUB PLAYM (at 80FB above) to play a note
809A 3B3 JNC      -0A 8090 ; Do again until end of data

809B 099 Y
809C 001 A
809D 004 D
809E 008 H
809F 014 T
80A0 012 R
80A1 002 B
80A2 379 *          ; BRTHDAY: FOCAL entry point (put in the ROM FAT)
80A3 03C NCXQ      0FDE
80A4 090 *          ; GOSUB MCPLAY (at 8090 above) to play notes
80A5 072          ; Music data ...

80A6 070          80AF 0AB          80B8 001
80A7 083          80B0 09F          80B9 0BA
80A8 073          80B1 072          80BA 0B8
80A9 09B          80B2 070          80BB 0B3
80AA 097          80B3 0C3          80BC 09B
80AB 072          80B4 0B3          80BD 0AB
80AC 070          80B5 09B          80BE 09F
80AD 083          80B6 093          80BF 3FF
80AE 073          80B7 083          80C0 3E0 RTN

```

It should be noted, that BRTHDAY sounds a little like it is being played by the robots of the Sirius Cybernetics Corporation ... only slightly worse<sup>2</sup>. I am not sure that this is really a fitting tribute to the HP-41! If you want more, try these:

```

80C1 099 Y
80C2 012 R
80C3 012 R
80C4 005 E
80C5 00D M
80C6 379 *          ; MERRY: FOCAL entry point (put in the ROM FAT)
80C7 03C NCXQ      0FDE
80C8 090 *          ; GOSUB MCPLAY (at 8090 above) to play notes
80C9 0A1          ; Music data ...

80CA 091          80D3 0A1          80DC 0A3
80CB 081          80D4 0B1          80DD 091
80CC 091          80D5 0B3          80DE 091
80CD 0A1          80D6 0A1          80DF 0A1
80CE 0A1          80D7 091          80E0 091
80CF 0A3          80D8 081          80E1 084
80D0 091          80D9 091          80E2 3FF
80D1 091          80DA 0A1          80E3 3E0 RTN
80D2 093          80DB 0A1

```

<sup>2</sup> <https://youtu.be/dM-t3I5uFro>

```

8171 085 E
8172 00C L
8173 007 G
8174 00E N
8175 009 I
8176 00A J
8177 379 * ; JINGLE: FOCAL entry point (put in the ROM FAT)
8178 03C NCXQ 0FDE ; GOSUB MCPLAY (at 8090 above) to play notes
8179 090 * ; Music data ...
817A 0A1

817B 0A1 818C 0A1 819D 0A9
817C 0A3 818D 0A1 819E 0A9
817D 0A1 818E 091 819F 0A9
817E 0A1 818F 091 81A0 0A9
817F 0A3 8190 0A1 81A1 0A9
8180 0A1 8191 096 81A2 0A1
8181 0B1 8192 0A1 81A3 0A1
8182 082 8193 0A1 81A4 0A1
8183 090 8194 0A3 81A5 0B1
8184 0A6 8195 0A1 81A6 0B1
8185 0A9 8196 0A1 81A7 0A9
8186 0A9 8197 0A3 81A8 091
8187 0A9 8198 0A1 81A9 085
8188 0A9 8199 0B1 81AA 3FF
8189 0A9 819A 082 81AB 3E0 RTN
818A 0A1 819B 090
818B 0A1 819C 0A6

81AC 092 R
81AD 001 A
81AE 014 T
81AF 013 S
81B0 379 * ; STAR: FOCAL entry point (put in the ROM FAT)
81B1 03C NCXQ 0FDE ; GOSUB MCPLAY (at 8090 above) to play notes
81B2 090 * ; Music data ...
81B3 099

81B4 099 81C3 0B9 81D2 0C1
81B5 0C1 81C4 0B9 81D3 0C9
81B6 0C1 81C5 0B1 81D4 0C9
81B7 0C9 81C6 0B1 81D5 0C3
81B8 0C9 81C7 0AB 81D6 0B9
81B9 0C3 81C8 0C1 81D7 0B9
81BA 0B9 81C9 0C1 81D8 0B1
81BB 0B9 81CA 0B9 81D9 0B1
81BC 0B1 81CB 0B9 81DA 0A9
81BD 0B1 81CC 0B1 81DB 0A9
81BE 0A9 81CD 0B1 81DC 09B
81BF 0A9 81CE 0AB 81DD 3FF
81C0 09B 81CF 099 81DE 3E0 RTN
81C1 0C1 81D0 099
81C2 0C1 81D1 0C1

```

To save you typing in the code, the complete PLAY ROM is available on the HP-41 page at [hpcc.org](http://hpcc.org). The ROM also includes my MCODE Metronome which was featured in Datafile V9N4P19. More beeper control functions appear in V7N7P12.

*.END.*